
지금까지 액션스크립트 인터프리터에서 실행시키기 위한 명령어를 만드는 법에

대해자세히살펴보았다. 이제인터프리터에‘어떤’일을할지알려주는것은꽤익

숙해졌을 것이다. 그런데‘언제’그러한 일을 할지는 어떻게 알려줄 수 있을까? 액

션스크립트 코드는 저절로 실행되지 않는다. 무엇인가 그 코드를 실행하도록 원인

을제공해야한다.

그‘무엇’은무비를동기화시켜서순서대로재생하는것일수도있고미리정해

진비동기화된이벤트일수도있다.

동기 코드 실행
무비를재생하면타임라인의플레이헤드가프레임에서프레임으로움직인다. 플

레이헤드가새프레임으로들어갈때마다인터프리터에서는그프레임에있는코드

를 실행시킨다. 프레임에 있는 코드가 실행되면 화면에 표시되는 내용이 업데이트

되고소리도재생된다. 그리고나면플레이헤드가다음프레임으로이동한다.

예를들어무비의 1번프레임에코드를직접넣으면그코드는 1번프레임에있

는 내용이 화면에 표시되기 전에 실행된다. 같은 무비의 5번 프레임에 있는 키프레

임에 다른 코드 블록을 넣으면 1번부터 4번 프레임까지 재생한 다음 5번 프레임에

있는코드를실행하고, 그 다음에 5번프레임이화면에표시된다. 1번프레임과 5번

프레임에서 실행된 코드는‘동기화되어 실행된다’고 할 수 있다. 선형적이고 예측

할수있는순서대로실행되기때문이다.

무비의 프레임에 들어 있는 코드는 모두 동기화되어 실행된다. gotoAndPlay()

나 gotoAndStop() 명령어 때문에 프레임 번호 순서와 다른 순서로 재생되는 프레

임이있더라도각프레임에있는코드는예측할수있는순서로플레이헤드와동기

화되어실행된다.

이벤트 기반의 비동기 코드 실행
어떤코드는실행되는순서를예측할수없는것도있다. 대신그러한코드는액

션스크립트 인터프리터에서 미리 정해진‘이벤트(event)’가 일어났다는 것을 알게

되면 실행된다. 많은 이벤트는 마우스를 클릭하거나 키를 누르는 것과 같은 사용자

행동과 연관되어 있다. 플레이헤드가 새로운 프레임에 들어가면 프레임 코드가 동

기화되어실행되는것처럼이벤트는이벤트가발생하면이벤트기반코드가실행된

다. 이벤트가언제든일어날수있기때문에이벤트기반코드(이벤트가발생하면실

행되는코드)는‘비동기적으로실행된다’라고말한다.

동기화된 프로그래밍을 하려면 코드가 언제 실행될지 미리 알 수 있어야 한다.

반면에비동기프로그래밍을이용하면이벤트가일어날때마다동적으로반응할수

있다. 비동기 코드를 실행할 수 있다는 점은 액션스크립트와 플래시의 상호작용에

서매우결정적인요소이다.

이 장에서는 플래시에서의 비동기 (이벤트 기반) 프로그래밍과 액션스크립트에

서지원하는다양한이벤트에대해알아보자.

이벤트 유형
이벤트는크게두가지범주로나눌수있다.

사용자 이벤트

사용자행동(마우스클릭이나키를누르는것)

시스템 이벤트

무비를 내부적으로 재생하는 과정에서 일어나는 일(예: 무비 클립이 스테이지에

나타나는경우또는외부파일로부터일련의변수를로딩하는경우)

액션스크립트에서는사용자이벤트와시스템이벤트를정확히구분하지않는다.

무비에 의해 내부적으로 일어나는 이벤트라고 해서 사용자가 마우스를 클릭할 때

일어나는 이벤트와 다르지는 않다. 보통 어떤 무비 클립이 스테이지에서 없어지는

것이 그다지 대단한 이벤트로 느껴지진 않겠지만 시스템 이벤트에 반응할 수 있다

는것은무비를제어하는데크게도움이된다.

액션스크립트 이벤트는 그 이벤트가 속하는 객체를 기준으로 나눌 수도 있다.

모든 이벤트는 플래시 환경에서 어떤 객체와 연관되어 발생한다. 즉 인터프리터에

서는‘사용자가 클릭했다’라고 하지는 않는다. 대신‘사용자가 이 버튼을 클릭했습

니다’또는‘이무비클립이스테이지에있을때사용자가마우스를클릭했습니다’

라고말한다. 또한‘데이터를받았습니다’가아니라‘이무비클립에서어떤데이터

를받았습니다’라고말한다. 어떤이벤트에반응하기위한코드를정의할때는이벤

트와관련된객체에서정의하게된다.

이벤트를받아들일수있는액션스크립트객체는다음과같다.

䤎 무비클립

䤎 버튼

䤎 XML 및 XMLSocket 클래스의객체

이 장 전반에 걸쳐 배우게 되겠지만, 액션스크립트에서는 사실 두 종류의 서로

다른 이벤트를 구현한다. 하나는 무비 클립 및 버튼과 연관된 이벤트이고 나머지는

다른모든종류의객체와연관된이벤트이다.

이벤트 핸들러
모든이벤트에서코드를실행하도록만드는것은아니다. 이벤트는무비에어떤

영향을미치지않고도정기적으로일어난다. 예를들어어떤사용자가한버튼을계

속 클릭하여 수십개의 이벤트를 발생시킬 수도 있지만, 그러한 클릭을 무시해버릴

수도 있다. 어떻게 그렇게 할 수 있을까? 이벤트가 일어나는 것만으로는 어떤 코드

를실행시킬수없고이벤트에반응하는코드를별도로만들어야하기때문이다. 어

떤 이벤트가 일어났을 때 인터프리터에서 어떤 코드를 실행하도록 하려면 특정 이

벤트가일어날때해야할일을기술하는‘이벤트핸들러’라는것을추가해야한다.

이벤트핸들러라는이름이붙은이유는이벤트핸들러에서무비에서일어난이벤트

를‘잡거나(catch)’‘처리(handle)’하기때문이다.

이벤트핸들러는특정이벤트가발생할때자동으로호출되는특별한이름을가

진함수라고볼수있다. 따라서이벤트핸들러를만드는것은몇가지차이점을제

외하면함수를만드는것과매우흡사하다.

䤎 이벤트 핸들러에는 keyDown과 같이 미리 정해진 이름이 있다. 이벤트 핸

들러의 이름은 프로그래머가 마음대로 정할수 있는 것이아니라 [표 10-1]

과 [표 10-2]에나온것처럼미리정해진이름을사용해야한다.

䤎 이벤트핸들러는 function 선언문으로선언하지않는다.

䤎 이벤트핸들러는버튼, 무비클립또는객체에집어넣을수있으며프레임에

서는사용할수없다.

이벤트 핸들러 문법
액션스크립트 이벤트 이름(또한 그에 상응하는 이벤트 핸들러의 이름)은 모두 미

리정해져있다. 버튼 이벤트핸들러는 on (eventName)을 이용하여정의하고무비

클립 이벤트 핸들러는 onClipEvent (eventName)을 이용하여 정의한다. 여기서

eventName은처리할이벤트의이름이다.

따라서 모든 버튼 이벤트 핸들러(key 매개변수가 필요한 keyPress는 제외)는 다

음과같은형식으로쓰인다.

on (eventName) {

statements

}

하나의버튼이벤트핸들러로여러개의이벤트를처리할수도있다. 이 때는각

이벤트를쉼표로구분한다. 예를들면다음과같다.

on (rollover, rollOut) {

// rollOver와 rollOut 이벤트가 발생하면 실행할 사용자 정의 함수를 호출한다.

playRandomSound();

}

모든무비클립이벤트핸들러는다음과같은형식으로쓰인다.

onClipEvent (eventName) {

statements

}

버튼이벤트핸들러와는달리클립이벤트핸들러는하나의이벤트만처리할수

있다.

이벤트 핸들러 만들기
이벤트핸들러를만들때는핸들러를만들고그핸들러를적절한객체에집어넣

으면 된다. 우선 가장 일반적인 핸들러인 버튼 이벤트 핸들러와 무비 클립 이벤트

핸들러에대해알아보자.

버튼과 무비 클립에 이벤트 핸들러 추가하기

버튼이나 무비 클립에 이벤트 핸들러를 추가하려면 핸들러 함수 코드를 원하는

버튼 또는 클립에 추가해야 한다. 이러한 작업을 하려면 [그림 10-1]에 나온 것과

같이플래시저작도구에서스테이지에있는객체를선택하고 Actions 패널에서적

절한코드를입력해야한다.

버튼과 무비 클립에 간단한 이벤트 핸들러 함수를 추가해 보자. 우선 버튼 이벤

트핸들러를만들려면다음과같이하면된다.

1. 새로운플래시무비를시작한다.

2. 버튼을만들고그버튼의인스턴스를메인스테이지에끌어다놓는다.

3. 버튼을선택한후 Actions 패널에다음과같은코드를입력한다.

on (release) {

trace("You clicked the button");

}

4. Control → Test Movie를선택한다.

5. 버튼을클릭한다. 그러면“You clicked the button”이라는메시지가 Output

창에출력된다.

무비가 재생되고 버튼을 눌렀다 떼면 인터프리터에서는 release 이벤트를 감지

하고 on (release) 이벤트 핸들러를 실행한다. 매번 버튼을 눌렀다 뗄 때마다

Output 창에“You clicked the button”이라는메시지가출력된다.

이제 무비 클립에 조금 다른 이벤트 핸들러를 추가해 보자. 이번에는 다음 과정

을따라해보자.

1. 새로운플래시무비를시작한다.

2. 메인무비스테이지에직사각형을그린다.

3. Insert → Convert to Symbol을선택한다.

4. Symbol Properties 대화상자에서새로운심볼의이름을 rectangle이라고입

력하고 Behavior에서 Movie Clip을선택한다.

5. OK를클릭하여 rectangle 무비클립을만드는작업을끝낸다.

6. 스테이지에서 rectangle 클립을선택한후다음과같은내용을 Actions 패널

에입력한다.

onClipEvent (keyDown) {

_visible = 0;

}

onClipEvent (keyUp) {

_visible = 1;

}

7. Control → Test Movie를선택한다.

8. 무비를 클릭하여 키보드 포커스가 그 무비에 있는지 확인한다. 그리고 나서

키를 누른채 기다려 보자. 키를 누르면 rectangle 무비 클립이 사라고 눌렀

던키에서손을떼면 rectangle 클립이다시나타난다.

핸들러를호출하는선언문은별도로만들필요가없다는점에주의하자. 핸들러

와연관된이벤트가발생하면인터프리터에서자동으로이벤트핸들러를호출한다.

플래시에서는 무비를 실행하는 도중에 액션스크립트를 통해 핸들러를 추가하거

나 제거하는 기능을 지원하지 않는다. 이벤트 핸들러는 플래시 저작도구를 이용하

여버튼이나클립에할당해야한다. 따라서다음과같은구문은사용할수없다.

myClip.onKeyDown = function () { _visible = 0; };

이러한단점을극복하는방법은‘동적무비클립이벤트핸들러’절에서알아보

기로하자.

다른 객체에 이벤트 핸들러 추가하기

무비 클립이나 버튼 외에도 두 개의 내장 객체 클래스(XML과 XMLSocket)에서

이벤트 핸들러를 지원한다. 이러한 객체에서는 이벤트 핸들러가 저작 도구에 있는

어떤 물리적인 대상에 포함되지 않는다. 대신 이벤트 핸들러가 객체 인스턴스에 대

한메소드로들어간다.

액션스크립트에서는 XML과 XMLSocket 객체에서이벤트핸들러로사용하기위

한 속성 이름이 미리 정해져 있다. 예를 들어 onLoad 속성은 외부 XML 데이터를

가져왔을때실행할핸들러의이름이된다.

XML 객체에대한 onLoad 속성을설정하려면다음과같은코드를사용하면된다.

myDoc = new XML();

myDoc.onLoad = function () { trace("all done loading!"); };

또는 핸들러 함수를 먼저 정의한 다음 그 함수를 객체의 onLoad 속성에 대입할

수도있다.

function doneMsg () {

trace("all done loading!");

}

myDoc.onLoad = doneMsg;

이런 구문은 [예제 10-1]에 나온 것처럼 이벤트 핸들러 속성에 함수를 대입할

수있는자바스크립트의경우와비슷하다.

[예제 10-1] 자바스크립트 이벤트 핸들러 대입

// 자바스크립트의 onload 핸들러에 함수 리터럴을 대입한다.

window.onload = function () { alert("done loading"); };

// 또는 함수를 만들어서 onload 속성에 대입해도 된다.

function doneMsg () {

alert("done loading");

}

window.onload = doneMsg;

플래시 차기 버전에서는 더 많은 액션스크립트 객체에서 객체 속성을 이용하여

이벤트 핸들러를 할당하는 방식을 지원할 가능성이 있으므로, 이러한 스타일에 미

리 익숙해지는 것이 좋다. XML이나 XMLSocket 객체를 사용하지 않더라도 HTML

문서와 자바스크립트를 이용하여 이러한 방식으로 핸들러 만드는 연습을 해 볼 수

도있다. 이런방법은활용범위가매우넓다는장점을가지고있다. 이렇게하면무

비를 재생하는 도중에 어떤 이벤트 핸들러 함수라도 손쉽게 새로 할당하거나 제거

할수있다.

객체에 함수를 추가하는 방법은‘12장. 객체와 클래스’에서 더 자세하게 배울

것이다. XML과 XMLSocket 객체에서지원하는이벤트에대한정보는‘3부. 레퍼런

스’에서찾아볼수있다.

이벤트 핸들러 영역
다른함수에있는선언문과마찬가지로이벤트핸들러에있는선언문은미리정

해진 영역 안에서 실행된다. 영역에 의해 인터프리터가 이벤트 핸들러에서 참조한

변수, 하위 함수, 객체, 속성 등을결정하는범위가정해진다. 무비 클립이벤트, 버

튼이벤트, 다른객체이벤트에대한이벤트핸들러영역을각각알아보도록하자.

이벤트 핸들러의 생존 기간은 그 이벤트 핸들러가 속해 있는 객체의 생존 기간에 의해 결정된
다. 클립이나 버튼이 스테이지에서 사라지거나 XML 또는 XMLSocket 객체가 없어지면 그와 관
련된 이벤트 핸들러도 같이 제거된다. 핸들러가 동작하려면 객체가 스테이지에 있거나 타임
라인에 존재해야 한다.



무비 클립 이벤트 핸들러 영역

보통함수와는달리무비클립이벤트핸들러에는지역영역이라는것이정의되

어있지않다. 클립에핸들러를추가하면핸들러영역은이벤트핸들러자체에국한

되는것이아니라클립전체가된다. 즉 모든변수를클립의타임라인으로부터알아

낸다. 예를 들면 navigation이라는 클립에 enterFrame 이벤트 핸들러를 추가하고

그 핸들러에 trace(x);라는 선언문을 입력하면, 인터프리터에서는 x 값을

navigation의타임라인에서검색한다.

onClipEvent (enterFrame) {

trace(x);  // navigation.x의 값이 출력된다.

}

검색할만한지역영역이아예없기때문에, 인터프리터에서는지역영역을먼저

검색하지 않는다. 핸들러 안에서 var y = 10;이라는 선언문을 실행시켜도 y는

navigation의타임라인에서정의된다(일반적인함수에서 var 키워드를사용하면함수

안에서만쓸수있는지역변수를선언하게된다).

핸들러 선언문을 핸들러 클립의 프레임에 포함된 것처럼 생각하면 클립 이벤트

핸들러의 영역을 따지는 규칙을 기억하기에 편리하다. 예를 들어 xVelocity라는 변

수를 포함하고 있는 ball이라는 클립이 있다고 생각해 보자. ball의 이벤트 핸들러

내부에서 xVelocity를사용할때는다음과같이그냥변수이름만사용하면된다.

onClipEvent (mouseDown) {

xVelocity += 10;

}

인터프리터에서이미우리가 ball 클립에있는 xVelocity라는변수를사용하려고

한다는 것을 알고 있기 때문에, _root.ball.xVelocity와 같이 변수의 경로까지 적어

줄필요는없다. 속성과메소드에대해서도똑같은규칙이적용된다. ball._x가아니

라 _x만, ball.gotoAndStop(5)가 아니라 gotoAndStop(5)만 적어주면 된다. 예를

들면다음과같다.

onClipEvent (enterFrame) {

_x += xVelocity;                 // 공을 움직인다.

gotoAndPlay(_currentframe - 1);  // 간단한 루프 반복
}

심지어다음과같이핸들러에서 function 선언문을이용하여 ball에포함되는함

수를정의할수도있다.

onClipEvent (load) {

function hideMe() {

_visibility = 0;

}

}

클립이벤트핸들러에있는선언문의영역은핸들러함수의지역영역이나클립

의 부모 타임라인(클립이 있는 곳 바로 위의 타임라인)이 아니라 클립의 타임라인이

라는점을잊어버리는경우도종종있다.

예를들어 ball 클립을무비의메인타임라인에놓고 (ball의타임라인이아닌) 메

인 타임라인에서 moveBall()이라는 함수를 정의했다고 하자. 별 생각 없이 ball에

있는이벤트핸들러에서다음과같이 moveBall()을호출할수도있다.

onClipEvent (enterFrame) {

moveBall();  // ball에는 moveBall()이라는 함수가 없으므로
// 아무 일도 일어나지 않는다. 

// moveBall() 함수는 _root에서 정의한 함수이다.

}

이런 경우에는 다음과 같이 _root를 이용하여 moveBall() 함수가 메인 타임라

인에있다는것을알려줘야한다.

onClipEvent (enterFrame) {

_root.moveBall();  // 이렇게 하면 제대로 동작한다.

}

때에 따라 이벤트 핸들러에서 현재 클립 객체를 직접 참조해야 하는 경우도 있

다. 이런경우에는 this 키워드를사용하면되는데, 이 키워드는이벤트핸들러안에

서현재무비클립을가리키는역할을한다. 따라서아래와같은레퍼런스는이벤트

핸들러내부에서쓰면똑같은변수를가리킨다.

this._x  // _x와 똑같다.

_x

this.gotoAndStop(12);  // gotoAndStop(12)와 같다.

gotoAndStop(12);

현재클립의속성(변수 또는 클립) 이름을동적으로생성할때 this를가장많이

사용한다. 아래 코드에서는 ball 클립의 현재 프레임 번호에 따라 ball.stripe1,

ball.stripe2, ...와같은이름을가진클립중하나를재생한다.

onClipEvent (enterFrame) {

this["stripe" + _currentframe].play();

}

this라는 키워드는 무비 클립 객체에 대한 직접적인 레퍼런스가 필요한 무비 클

립 메소드에서도 자주 쓰인다. 액션스크립트의 전역 함수와 같은 이름을 가진 무비

클립메소드를사용할때는직접클립레퍼런스를적어줘야한다. 따라서다음과같

은 함수를 이벤트 핸들러 내부에서 메소드로 호출할 때는 this 키워드를 사용해야

한다.

duplicateMovieClip()

loadMovie()

loadVariables()

print()

printAsBitmap()

removeMovieClip()

startDrag()

unloadMovie()

예를들면다음과같다.

this.duplicateMovieClip("ball2", 1);

this.loadVariables("vars.txt");

this.startDrag(true);

this.unloadMovie();

이처럼 이름이 같은 함수 때문에 생기는 문제는‘13장. 무비 클립’의‘메소드와

전역함수가겹치는문제’라는절에서조금더알아보기로하자.

this 키워드를이용하면저작도구에서현재클립의이름을정해주지않았거나클

립 이름을 모르는 경우에도 현재 클립을 참조할 수 있다는 점을 기억해 두자. 게다

가 현재 클립의 이름을 모르는 경우에도 this 키워드만 사용하면 현재 클립을 다른

함수에레퍼런스로전달할수있다. 아래예제에이러한기능을사용하여만든코드

가나와있다.

// 메메인인 타타임임라라인인의의 코코드드
// 어떤 클립이든 움직일 수 있는 범용 함수
function move (clip, x, y) {

clip._x += x;

clip._y += y;

}

// 클클립립에에 들들어어가가는는 코코드드
// 메인 타임라인에 있는 함수를 호출하여 현재 클립을 움직인다.

// 이 때 this 키워드를 이용하여 현재 클립에 대한 레퍼런스를 전달한다.

onClipEvent (enterFrame) {

_root.move(this, 10, 15);

}

버튼 이벤트 핸들러 영역

버튼 핸들러 영역은 그 버튼이 속해 있는 타임라인 영역에 들어간다. 예를 들어

어떤버튼을메인타임라인에추가하고그버튼에있는핸들러에서 speed라는변수

를선언하면 speed 영역은메인타임라인(_root)에속하게된다.

플래시 5의 빌드 30에는 버그가 있어서 gotoAndStop()과 gotoAndPlay() 함수를 클립 핸들
러 안에서 실행시킬 때 문자열 리터럴 레이블을 사용할 수 없다. 그러한 명령이 나오면 그
냥 무시된다. 예를 들면 다음과 같은 코드는 작동하지 않는다.

onClipEvent(load) {

gotoAndStop("intro");  // 플래시 5 r30에서는작동하지않는다.

}

이러한 버그를 피하려면 다음과 같은 클립 레퍼런스를 사용하면 된다.

onClipEvent(load) {

this.gotoAndStop("intro");

}



// 버버튼튼 핸핸들들러러에에 들들어어가가는는 코코드드
on (release) {

var speed = 10;  // _root에서 speed라는 변수를 정의한다.

}

이와는 달리 메인 타임라인에 ball이라는 무비 클립을 놓고 ball의 핸들러에서

speed라는변수를선언하면 speed의영역은 ball에속하게된다.

// ball 핸핸들들러러에에 들들어어가가는는 코코드드
on (load) {

var speed = 10;  // _root가 아닌 ball에서 speed를 정의한다.

}

버튼 핸들러 안에서 this 키워드는 버튼을 포함하고 있는 타임라인을 가리키게

된다.

on (release) {

// 이 버튼을 포함하고 있는 클립의 투명도를 50%로 설정한다.

this._alpha = 50;

// 이 버튼을 포함하고 있는 클립을 오른쪽으로 10 픽셀 옮긴다.

this._x += 10;

}

다른 객체 이벤트 핸들러 영역

무비클립핸들러또는버튼핸들러와달리 XML이나 XMLSocket과같은내장클

래스의 인스턴스에 들어가는 이벤트 핸들러 영역은 함수 영역과 똑같다. XML 또는

XMLSocket 객체의 이벤트 핸들러의 영역 사슬은 핸들러 함수를 정의할 때 함께 정

의되는영역사슬과같다. 게다가 XML 및 XMLSocket 이벤트핸들러에는지역영역

도있다. ‘9장. 함수’의‘함수영역’절에서설명한모든함수영역규칙은버튼또는

무비클립을제외한객체에들어가는이벤트핸들러함수에도그대로적용된다.

버튼 이벤트
[표 10-1]에 버튼에서 사용할 수 있는 다양한 이벤트를 간단하게 정리해 놓았

다. 버튼 이벤트를이용하면다른위치로의이동, 폼, 게임 외의다양한인터페이스

요소에서사용할수있는코드를손쉽게만들수있다. 각 버튼이벤트에대해알아

보고 버튼에서 마우스나 키보드 이벤트에 반응할 수 있도록 프로그램을 짜는 방법

을알아보기로하자.

[표 10-1]에 나온 각 버튼 이벤트는 on (eventName) 형태로 주어지는, 각 이벤

트에 상응하는 버튼 이벤트 핸들러에서 처리한다. 예를 들어 press 이벤트는 on

(press)로 시작하는 이벤트 핸들러로 처리한다. 단 한 가지 예외가 있다. keyPress

이벤트 핸들러는 on (keyPress key)와 같은 형식으로 시작되는데, 이 때 key에는

그 이벤트 핸들러에서 처리할 키가 들어간다. 버튼 이벤트는 마우스와 상호작용을

하는 버튼에만 전달된다. 여러 개의 버튼이 서로 겹쳐 있는 경우에는 가장 위에 있

는버튼에서모든이벤트를받아들이며, 가장위에있는버튼에이벤트핸들러가없

는경우에도다른버튼에서는주어진이벤트에반응할수없다. 이 책에서는버튼이

활성화되기 위해 마우스포인터가있어야할버튼영역을 가리킬때‘히트영역(hit

area)’이라는용어를사용하겠다(버튼의히트영역은플래시저작도구에서버튼을만

들때그래픽작업에의해정의된다).

[표 10-1] 버튼 이벤트

버튼 이벤트 이름 버튼 이벤트가 일어나는 경우

press 포인터가 버튼의 히트 영역에 있을 때 마우스의 주 버튼을 누른 경우.
다른 마우스 버튼은 무시된다.

release 포인터가 버튼의 히트 영역에 있을 때 마우스의 주 버튼을 눌렀다가 떼
는 경우

releaseOutside 포인터가 버튼의 히트 영역에 있을 때는 마우스의 주 버튼을 누르고 있
다가 포인터가 히트 영역에서 벗어나면 버튼에서 손을 떼는 경우

rollOver 마우스 버튼을 누르지 않은 상태에서 마우스 포인터를 버튼의 히트 영
역으로 이동시킨 경우

rollOut 마우스 버튼을 누르지 않은 상태에서 마우스 포인터를 버튼의 히트 영
역 밖으로 이동시킨 경우

dragOut 포인터가 버튼의 히트 영역 안에 있는 상태에서 마우스의 주 버튼을 누
르고 마우스 버튼을 누른 채로 포인터를 히트 영역 밖으로 움직인 경우

버튼 이벤트 이름 버튼 이벤트가 일어나는 경우

dragOver 포인터가 버튼의 히트 영역 안에 있는 상태에서 마우스의 주 버튼을 누
르고 마우스 버튼을 누른 채로 포인터를 히트 영역 밖으로 움직였다가
다시 히트 영역 안으로 끌고 오는 경우

keyPress 주어진 키를 누른 경우. 대부분의 경우에 keyPress 버튼 이벤트보다는
keyDown 클립 이벤트를 주로 사용한다.

press

마우스클릭을자세히보면두단계로진행된다는점을알수있다. 우선마우스

버튼을누르는(press) 단계와버튼에서손가락을떼는(release) 단계로나눌수있기

때문이다. press 이벤트는마우스포인터가히트영역안에있는상태에서마우스의

주버튼1)을누르는경우에발생한다. 다른마우스버튼은아무리눌러도 press 이벤

트가발생하지않는다. 버튼 press 이벤트는라디오버튼이나게임에서의무기발사

버튼과 같은 용도에 적합하지만, 마우스 버튼에서 손가락을 떼기 전에 사용자가 생

각이바뀌면선택을취소할수있도록하고싶다면 press 대신 release 이벤트를이

용하는것이좋다.

release

release 버튼이벤트는다음과같은일련의동작이일어날때발생한다.

1. 마우스포인터가버튼의히트영역안에있다.

2. 마우스 포인터가 버튼의히트영역안에있는상태에서 마우스의주버튼을

누른다(이때 press 이벤트가발생한다).

3. 마우스 포인터가 여전히원래버튼의 히트영역안에있는상태에서마우스

의주버튼에서손가락을뗀다(이순간에 release 이벤트가발생한다).

press 이벤트대신 release 이벤트를사용하는경우에는사용자가마우스버튼을

누른후에도포인터를버튼밖으로옮기고나서손가락을떼면 release 이벤트가발

생하지않으므로사용자가처음에내렸던결정을취소할수있다.

releaseOutside

releaseOutside 이벤트는보통사용자가버튼을클릭했다가마우스버튼에서손

가락을 떼기 전에 포인터를 버튼 밖으로 움직여서 버튼을 선택하려다가 만 경우에

발생한다. 이이벤트는다음과같은일련의동작이일어나면발생한다.

1. 마우스포인터가버튼의히트영역안에있다.

2. 마우스의 주 버튼을 누르고 손가락을 버튼에서 떼지 않는다(이 때 press 이

벤트가발생한다).

3. 마우스포인터를버튼의히트영역밖으로움직인다(이때 dragOut 이벤트가

발생한다).

4. 원래버튼의히트영역밖에서마우스의주버튼에서손가락을뗀다.

굳이 releaseOut 이벤트를이용하는경우는거의없을것이다. 사용자가버튼을

클릭했다가마음이바뀌어서취소하려고하는경우에보통 releaseOut 이벤트가발

생하기때문이다.

rollOver

rollOver 이벤트는 마우스 버튼을 누르지 않은 상태에서 마우스 포인터를 히트

영역 안으로 움직일 때 발생한다. 포인터가 버튼 위로 움직일 때 버튼 모양을 바꾸

거나 하는 작업은 스크립트를 이용하지 않고 플래시 저작 도구에서 직접 처리하기

때문에액션스크립트에서 rollOver 이벤트를사용하는일은거의없다. 버튼을강조

하는 기능을 구현할 때는 저작 도구에서 제공하는 up, over, down 프레임을 이용

해야한다.

플래시 5의 rollOver 이벤트를이용하면텍스트필드선택사항을손쉽게읽어들

일수있다. 자세한내용은 3부의‘선택객체’를참조하기바란다.

rollOut

rollOut 이벤트는 rollOver 이벤트와반대되는경우에발생한다. 즉마우스버튼을

누르지않은상태에서마우스포인터를히트영역밖으로움직이는경우에 rollOut 이

벤트가발생한다. 버튼강조상태는저작도구에서직접만들기때문에액션스크립트

에서별도로이미지를바꿀필요가없으므로 rollOver와마찬가지로 rollOut 이벤트도

거의쓰이지않는다.

dragOut

dragOut 이벤트는 포인터가 버튼의 히트 영역에서 벗어날 때 마우스 버튼이 눌

려 있는 상태에서 발생한다는 점을 제외하면 rollOut 이벤트와 거의 비슷하다.

dragOut 이벤트뒤에는반드시 releaseOutside 이벤트(사용자가마우스버튼에서손

가락을 떼는 경우)나 dragOver 이벤트(사용자가 마우스 버튼을 누른 채로 마우스 포

인터를버튼의히트영역으로다시끌고가는경우)가뒤따른다.

dragOver

dragOver 이벤트는 그다지 자주 일어나는 이벤트는 아니다. 이 이벤트는 다음

과같은일련의동작에의해발생한다.

1. 마우스 포인터가 버튼의 히트 영역 안으로 들어간다(rollOver 이벤트가 발생

한다).

2. 마우스의주버튼을누른채로손가락을떼지않는다(press 이벤트가발생한다).

3. 마우스 포인터를 버튼의 히트 영역 밖으로 움직인다(dragOut 이벤트가 발생

한다).

4. 마우스 포인터를 버튼의 히트 영역 안으로 다시 움직인다(dragOver 이벤트

가발생한다).

따라서 dragOver 이벤트는사용자가마우스버튼을계속누른상태에서마우스

포인터를 히트 영역 밖으로 움직였다가 다시 히트 영역 안쪽으로 움직였다는 것을

의미한다. 마우스 포인터가 버튼의 히트 영역에서 나왔다가 다시 그 안으로 들어갈

때 마우스 버튼을 누른 상태이면 rollOver 이벤트가 아닌 dragOver 이벤트가 발생

한다는점에주의하자.

keyPress

keyPress 이벤트는마우스이벤트와관련된것이아니라특정키를누르는것에

의해 생기는 이벤트이다. 이 이벤트를 지금 다루는 이유는 액션스크립트의 다른 버

튼 이벤트 핸들러와 마찬가지로 on (eventName)과 같은 구조를 가지기 때문이다.

이이벤트핸들러에서는그이벤트를발생시키는키를지정해주어야한다.

on (keyPress key) {

statements

}

여기서 key는그이벤트와관련된키를나타내는문자열이다. 이 문자열에는키

에 나와 있는 글자(“s”나“S”등) 또는“<Keyword>”와 같은 형식으로 표현된 키워

드가 들어갈 수 있다. 각 핸들러에서는 하나의 키밖에 지정할 수 없다. keyPress를

이용하여 여러 키를 감지하려면 다음과 같이 여러 개의 keyPress 이벤트 핸들러를

만들면된다.

// "a" 키를 감지한다.

on (keyPress "a") {

trace("The 'a' key was pressed");

}

// Enter 키를 감지한다.

on (keyPress "<Enter>") {

trace("The Enter key was pressed");

}

// 아래쪽 화살표 키를 감지한다.

on (keyPress "<Down>") {

trace("The Down Arrow key was pressed");

}

Keyword 자리에 사용할 수 있는 값들은 다음과 같다(F1부터 F12까지 기능키는

keyPress에서는지원하지않기때문에 Key 객체를이용해야한다).

<Backspace>

<Delete>

<Down>

<End>

<Enter>

<Home>

<Insert>

<Left>

<PgDn>

<PgUp>

<Right>

<Space>

<Tab>

<Up>

플래시 4에서는 keyPress를 이용하는 방법이 키보드를 감지할 수 있는 유일한

방법이었다. 플래시 5 이후에서는 Key 객체를 무비 클립 이벤트인 keyDown 및

keyUp(잠시 후에배울것이다)과 함께사용하여키보드를훨씬다양하게제어할수

있다. keyPress 이벤트를이용하는경우에는한번에하나의키를누르는것만감지

할 수 있지만, Key 객체를 이용하면 여러 개의 키를 한꺼번에 누르는 것도 감지할

수있다.

무비 클립 이벤트 개요
무비 클립 이벤트는 마우스 클릭에서 데이터 다운로드에 이르기까지 플래시 플

레이어 전반에 걸쳐 다양하게 일어난다. 클립 이벤트는 사용자 입력 이벤트와 무비

재생 이벤트의 두 가지 범주로 크게 나눌 수 있다. 사용자 입력 이벤트는 마우스나

키보드와관련되어있고무비재생이벤트는플래시플레이어에서프레임을렌더링

하는것이나무비클립의시작과끝, 데이터로딩따위와연관되어있다.

사용자 입력 클립 이벤트는 앞에서 다룬 버튼 이벤트와 그 기능 면에서 약간 겹

치는 부분이 있다. 예를 들어 어떤 클립의 mouseDown 이벤트 핸들러는 버튼의

press 이벤트 핸들러와 마찬가지로 마우스 버튼이 눌리는 것을 감지할 수 있다. 하

지만 무비클립이벤트는버튼이벤트처럼히트영역같은것에얽매여있지않으며

마우스포인터의모양에도영향을미치지않는다.

[표 10-2]에요약해놓은액션스크립트무비클립이벤트에대해자세히알아보자.

우선무비재생이벤트(enterFrame, load, unload, data)에대해배우고나서사용자입

력이벤트(mouseDown, mouseUp, mouseMove, keyDown, keyUp)에 대해 알아보기

로 하자. 각 클립 이벤트는 onClipEvent (eventName)과 같은 형식으로 시작하는 그

이벤트에상응하는클립이벤트핸들러에서처리한다. 예를 들어 enterFrame 이벤트

는 onClipEvent (enterFrame)으로시작하는이벤트핸들러로처리한다. load, unload,

data 이벤트를제외한모든무비클립이벤트는사용자가다른무비클립(또는무비클

립이아닌곳)을클릭하더라도스테이지에있는모든무비클립으로전달된다.

[표 10-1] 무비 클립 이벤트

클립 이벤트 이름 클립 이벤트가 발생하는 경우

enterFrame 플레이헤드가 프레임에 들어갈 때(플래시 플레이어에서 프레임을 렌더링
하기 전)

load 클립이 처음 스테이지에 등장할 때

unload 클립이 스테이지에서 사라질 때

data 클립의 변수 로딩이 완료되었을 때, 또는 로딩된 무비의 일부가 클립에 로
딩될 때

mouseDown 클립이 스테이지에 있는 상태에서 마우스의 주 버튼을 누를 때(다른 버튼
에는 반응하지 않음)

mouseUp 클립이 스테이지에 있는 상태에서 마우스의 주 버튼에서 손가락을 뗄 때

mouseMove 클립이 스테이지에 있는 상태에서 마우스 포인터가 조금이라도 움직일 때
(마우스가 클립 위에 있지 않아도 상관없음)

keyDown 클립이 스테이지에 있는 상태에서 키를 누를 때

keyUp 클립이 스테이지에 있는 상태에서 눌렀던 키에서 손가락을 뗄 때

무비 재생 무비 클립 이벤트
다음이벤트는플래시를로딩하고무비를재생하는과정에서사용자가아무것도

하지않더라도발생되는이벤트이다.

enterFrame

스크립트를 실행하기 위해 비어있는 무비로 루프를 돌려야 한다면, enterFrame

을 대신 사용할 수 있다. enterFrame 이벤트는 무비에서 각 프레임이 지나갈 때마

다 발생한다. 예를 들어 다음과 같은 코드를 무비 클립에 집어넣으면 그 클립은 프

레임마다 10픽셀씩커진다.

onClipEvent (enterFrame) {

_height += 10;

_width += 10;

}

(앞에서 배웠듯이 _height와 _width 속성은 enterFrame 이벤트 핸들러를 포함하고

있는클립영역에속하므로, _height와 _width 앞에클립인스턴스이름을붙이지않아

도된다)

어떤플래시무비라도플래시플레이어에서재생되면화면에서아무것도움직이

지않거나무비의플레이헤드가한프레임에서정지되어있더라도계속해서실행된

다. 개별 무비 클립의 enterFrame 핸들러는 그 클립이 재생 중이거나 정지해 있거

나상관없이클립이스테이지에있는한계속실행된다. gotoAndStop() 함수를 호

출하여 클립의 플레이헤드를 다른 곳으로 옮기더라도 각 프레임을 지날 때마다 클

립의 enterFrame 이벤트 핸들러는 여전히 실행된다. stop() 함수를 이용하여 무비

전체의 모든 플레이헤드가 멈추더라도 모든 클립에 있는 enterFrame 이벤트 핸들

러는계속해서실행된다.

enterFrame 이벤트는 일반적으로 무비 클립의 상태를 계속해서 업데이트하는

데사용된다. 하지만 enterFrame 이벤트핸들러가그핸들러를포함하고있는클립

에 직접 적용되어야 하는 것은 아니다. enterFrame을 코드를 반복 실행하기 위한

용도로 프레임이 하나뿐인 비어있는 클립에서 사용할 수도 있다. 클립 이벤트 루프

(또는 간단하게 프로세스라고도 부름)는‘8장. 순환문’의‘타임라인 루프와 클립 이

벤트루프’에서이미살펴보았다.

enterFrame 이벤트는 enterFrame 핸들러가 들어있는 클립의 플레이헤드가 멈추더라도 각 프
레임이 렌더링되기 전에 매번 발생한다. 따라서 enterFrame 이벤트는 계속해서 발생한다.



enterFrame 이벤트핸들러에있는코드는그핸들러를포함하고있는클립의타

임라인에들어있는다른어떤코드보다도먼저실행된다.

조금 더 욕심을 부리면 enterFrame을 이용하여 클립을 다양한 형태로 변화시킬

수도 있다. 뒤에 나오는 [예제 10-7]에서는 조금 전에 나왔던 클립 확대 코드를 확

장하여클립의크기가커졌다작아졌다하는기능을구현한다.

load

load 이벤트는 무비 클립이 탄생하는 순간(즉 무비 클립이 스테이지에 처음 나타

날때)에발생한다. 무비클립은다음중한가지방법을통해스테이지에등장한다.

䤎 플레이헤드가저작도구에서정해준위치에클립의새로운인스턴스를만들

어주는키프레임으로들어갈때

䤎 duplicateMovieClip() 함수를 통해 다른 클립으로부터 새로운 클립을 복사

할때

䤎 attachMovie() 함수를 이용하여 프로그래밍을 통해 클립을 스테이지에 추

가할때

䤎 loadMovie() 함수를이용하여외부의 .swf 파일을클립으로로딩할때

䤎 unloadMovie() 함수를 이용하여클립의내용을언로딩할때(클립의내용물

이사라지면그자리를메꾸기위한비어있는클립이클립으로로딩되기때문에

load 이벤트가발생한다)

load 이벤트핸들러의본체는무비클립이처음나타나는타임라인에있는다른

모든코드가실행된후에실행된다.

load 이벤트 핸들러는 클립의 변수를 초기화하거나 초기 설정 과정(동적으로 생

성된 클립의 크기나 위치를 조절하는 작업 등)을 처리하기 위한 목적으로 쓰이기도

한다. 또한 load 핸들러를이용하여무비클립이자동으로재생되는것을막을수도

있다.

onClipEvent (load) {

stop();

}

특정클립이있어야제대로실행되는함수를호출하는경우에도 load 이벤트핸

들러를활용할수있다.

load 이벤트는 새로운 무비 클립을 만드는 duplicateMovieClip() 함수와 함께

사용할 때 특히 유용하다. [예제 10-2]에서는 하나의 load 이벤트 핸들러를 연속적

으로 이용하여 star 클립으로 만들어진 화면을 만든다. load 핸들러는 star 클립을

복사할 때마다 함께 복사되므로 자기 자신을 계속해서 복사하게 된다. 100번째 클

립이 만들어지면 이 과정이 멈춘다. [예제 10-2]에서 쓰인 .fla 파일은 온라인 코드

창고에서구할수있다.

[예제 10-2] load 이벤트를 이용하여 별로 가득찬 화면 만들기

onClipEvent (load) {

// 현재 클립을 임의의 위치로 옮긴다.

_x = Math.floor(Math.random() * 550);

_y = Math.floor(Math.random() * 400);

// 이전 클립의 비율을 그대로 상속하지 않도록 클립 비율을 리셋한다.

_xscale = 100;

_yscale = 100;

// 50 퍼센트에서 150 퍼센트 사이의 임의의 값을 이용하여
// 현재 클립의 크기를 조절한다.

randScale = Math.floor(Math.random() * 100) - 50;

_xscale += randScale;

_yscale += randScale;

// 아직 100번째 별을 만들지 못했다면 또 다른 클립을 복사한다.

if (_name != "star100") {

nextStarNumber = number(_name.substring(4, _name.length)) + 1;

this.duplicateMovieClip("star" + nextStarNumber, nextStarNumber);

}

}

unload

unload 이벤트는 load 이벤트와 반대되는 이벤트로 무비 클립이 완료되면 발생

하는 이벤트이다. 즉 클립이 스테이지에 존재하는 마지막 프레임이 완료되자마자

(하지만다음프레임으로들어가기전에) 발생한다.

다음과같은상황에서무비클립의 unload 이벤트가발생한다.

䤎 플레이헤드가클립이존재하는기간의마지막프레임에올때

䤎 removeMovieClip() 함수(attachMovie() 또는 duplicateMovieClip() 함수로

만든클립을제거하는함수)로클립을제거했을때

䤎 unloadMovie() 함수를이용하여전에로딩했던 .swf 파일을제거했을때

䤎 클립에외부 .swf 파일을로딩했을때

마지막에나와있는상황은조금이상해보일지모르지만사실플래시에무비가

로딩되는 방법을 생각해 보면 당연히 그렇게 된다는 것을 알 수 있다. 무비 클립에

.swf 파일을 로딩하면 그 클립에 원래 들어있던 내용이 없어지므로 unload 이벤트

가발생하는것이다.

아래 예에서는 loadMovie() 함수와 연관된 load 및 unload 이벤트를 설명하고

있다.

1. 플래시저작도구에서무비의메인타임라인에있는 1번프레임의스테이지에

비어있는무비클립을추가한다. 그클립의이름을 emptyClip이라고하자.

2. 메인 타임라인의 5번 프레임에서 emptyClip.loadMovie(“test.swf”);라는 코

드를이용하여 test.swf라는무비를 emptyClip으로로딩한다.

3. Control → Play movie를선택하여무비를재생한다.

이렇게하면다음과같은결과가나온다.

1. 1번프레임: emptyClip 클립이나타나면서 load 이벤트가발생된다.

2. 5번프레임: 두단계에걸쳐 loadMovie() 함수가실행된다.

a. test.swf를 로딩할 수 있도록 emptyClip의 빈 자리를 차지하고 있던 내

용물을제거한다. 이때 unload 이벤트가발생한다.

b. test.swf라는무비가로딩되면서 load 이벤트가발생한다.

unload 이벤트는 뒷정리용 코드(스테이지를 정리하거나 프로그램 환경을 리셋하

는코드)를시작할때주로쓰인다. 또한 unload 핸들러를이용하면무비클립이끝

나고난뒤에어떤다른작업(다른무비를재생하는것등)을처리할수있다.

data

data 이벤트는무비클립에외부데이터가로딩되었을때발생한다. data 이벤트

는 로딩되는 데이터 종류에 따라 두 가지의 서로 상이한 상황에서 발생한다. 각 경

우를따로생각해보자.

loadVariables() 함수를 이용하여 서버에 일련의 변수를 요청하면 그 변수의 정

보를 활용하기 전에 우선 그러한 변수가 완전히 로딩될 때까지 기다려야 한다(3부

참조).

무비 클립에서로딩된변수의마지막부분을받고나면 data 이벤트가발생되어

로딩된값이필요한코드를실행해도좋다는것을알려준다.

예를들어방문객이의견을입력하고그러한의견을서버에저장하는방명록무

비가있다고생각해보자. 사용자가방문객이남긴의견을볼때는 loadVariables()

를 이용하여 서버에 그러한 정보를 요청한다. 하지만 사용자 의견을 화면에 표시하

기전에요청한데이터를모두사용할수있는지확인하기전까지는로딩화면을보

여준 채로 기다려야 한다. data 이벤트 핸들러를 이용하면 데이터 로딩이 완료되는

것을알수있기때문에방문객의의견을안전하게출력할수있다.

[예제 10-3]은 방명록 코드의 일부를 간단하게 정리해둔 것으로 loadVariables()

함수와함께 data 이벤트핸들러를이용하는방법을보여준다. 이예에서는한버튼에

서텍스트파일로부터두개의 URL로인코딩된변수를읽어들여무비클립으로로딩

한다. 이 무비 클립에는 변수가 모두 로딩된 후에 실행하는 data 이벤트 핸들러가 들

어있다. 핸들러에있는코드는 data 이벤트가일어나기전까지(즉 데이터를모두받기

전까지)는실행되지않으므로, 변수를화면에출력해도괜찮다는것을알수있다.

[예제 10-3] data 이벤트를 기다리는 예제

// guestbook.txt 파파일일의의 내내용용
name=judith&message=hello

// 클클립립에에 있있는는 버버튼튼
on (release) {

this.loadVariables("guestbook.txt");

}

// 클클립립에에 있있는는 핸핸들들러러
onClipEvent (data) {

trace(name);

trace(message);

}

‘17장. 플래시폼’에서플래시폼을만드는법에대해배울때 data 이벤트를조

금더자세히알아보자.

data 이벤트를 활용하는 두 번째 용도는 loadMovie() 함수를 이용하여 외부의

.swf 파일을무비클립으로로딩하는것과관련되어있다. .swf 파일을호스트클립

으로 로딩하면 파일을 모두 로딩하지 않은 경우에도 바로 파일이 재생된다. 하지만

이렇게 무조건 로딩하는 것이 항상 바람직한 것은 아니다. 재생을 시작하기 전에

.swf 파일을 모두 또는 일정 비율 이상 로딩해야 하는 경우도 있다. data 이벤트 핸

들러와프리로딩코드를이용하면그렇게만들수있다.

data 이벤트는호스트무비클립에서외부의 .swf 파일의일부를받을때마다발

생하는이벤트이다. 이 때‘일부’의의미는생각보다복잡하다. data 이벤트가발생

하려면, 마지막 data 이벤트가 발생한 후, 또는 .swf 파일 로딩을 시작한 후에 외부

.swf 파일 중 적어도 하나의 프레임이 로딩되어야 한다(그동안 .swf 파일로부터 두

개이상의프레임을로딩할수도있지만 data 이벤트가발생할수있는최소한의프레임

개수는한개다).

data 이벤트핸들러의실행은플레이어에서프레임을렌더링하는것과밀접하게

연관되어 있다. 각 프레임이 렌더링될 때마다 인터프리터에서는 로딩된 외부 .swf

파일의 일부분 가운데 data 이벤트 핸들러가 있는 클립에 로딩된 것이 있는지 확인

한다. 만약 외부 .swf 파일중일부가그러한클립에로딩되었다면, 그리고 새로 로

딩된 부분이 적어도 한 프레임 이상이라면 data 이벤트 핸들러가 실행된다. 이러한

과정은 (플레이헤드가멈추더라도) 한프레임을렌더링할때한번만발생한다.

data 이벤트는 매 프레임을 기준으로 발생하므로 프레임 속도가 빠른 무비일수

록프리로더의모양이더부드럽다. .swf 파일의로딩상황을더자주업데이트하기

때문이다.

loadMovie() 작업을 수행하는 중에 일어나는 data 이벤트의 정확한 개수는 로

딩하는 .swf 파일의내용분포와연결속도에따라달라진다. 한프레임짜리 .swf 파

일은 파일 크기가 아무리 커도 한 번에 하나의 data 이벤트만 발생시킨다. 반면에

100개의 프레임이 있는 .swf 파일은 무비의 프레임 속도나 각 프레임의 바이트 크

기, 네트워크연결속도등에따라달라질수있지만최대 100개의 data 이벤트까지

발생시킬 수 있다. 프레임이 크고 연결이 느릴 때 더 많은 data 이벤트가 발생된다

(최대 한 프레임에한번까지). 프레임이 작고연결속도가빠르면 data 이벤트횟수

가 줄어든다(플레이어에서 두 프레임을 렌더링하는 동안 100 프레임이 모두 전송되면

data 이벤트가한번밖에발생하지않는다).

어떻게 data 이벤트 핸들러를 이용하여 프리로더를 만들 수 있을까? load

Movie() 함수를 호출하여 data 이벤트가 발생되면 외부 .swf 파일을 다운로드하는

중이라는 것을 알 수 있다. 따라서 data 이벤트 핸들러 내부에서 파일을 재생하기

전에 파일이 충분히 다운로드되었는지 확인할 수 있다. 이러한 작업은 [예제 10-4]

에 나온 것처럼 getBytesLoaded()와 getBytesTotal() 함수를 이용하여 처리할 수

있다(무비클립속성인 _framesloaded와 _totalframes를이용해도된다).

[예제 10-4]에서는 무비를 로딩하는 중간에 다운로드 상황을 보여준다. 완전히

다운로드하기 전에 자동으로 재생되는 것을 방지하려면 로딩 중인 .swf 파일의 첫

번째프레임에서 stop() 함수를 호출해야 한다는 점에주의하자. 온라인 코드 창고

에서 [예제 10-4]를변형시킨코드를구할수있다.

[예제 10-4] data 이벤트 프리로더

onClipEvent (data) {

trace("data received");           // 프리로더 시작

// 데이터 전송을 나타내는 불을 켠다.

_root.transferIndicator.gotoAndStop("on");

// 로딩이 끝나면 데이터 전송을 나타내는 불을 끄고 무비를 재생한다.

if (getBytesTotal() > 0 && getBytesLoaded() == getBytesTotal()) {

_root.transferIndicator.gotoAndStop("off");

play();

}

// _root에 있는 텍스트 필드 변수에 로딩과 관련된 자세한 내용을 표시한다.

_root.bytesLoaded = getBytesLoaded();

_root.bytesTotal = getBytesTotal();

_root.clipURL = _url.substring(_url.lastIndexOf("/") + 1,

_url.length);

}

사용자 입력 무비 클립 이벤트
나머지무비클립이벤트는사용자와의상호작용과연관되어있다. 사용자입력

클립 이벤트가 발생하면 스테이지에 있는 모든 클립(다른 클립 안에 들어 있는 클립

도 상관없다)에서 그 이벤트를 받아들인다. 따라서 마우스를 클릭하거나 움직이거

나키보드의키를누를때, 한꺼번에여러개의클립이반응할수도있다.

특정 클립 근처에 마우스 포인터가 있을 때 반응하는 코드를 만들고 싶다면, 이

벤트핸들러에서클립에대한상대적인마우스포인터의위치를확인하도록만들면

된다. 잠시후에 [예제 10-9]에서볼수있겠지만 hitTest()라는내장함수를이용하

면마우스클릭이특정영역안에서일어났는지간단하게확인할수있다.

mouseDown

press 버튼 이벤트와 마찬가지로 mouseDown 이벤트는 마우스 버튼을 누르는

것을 감지한다. mouseDown 이벤트는 마우스 포인터가 스테이지의 어느 위치에

있더라도마우스의주버튼을누를때마다발생한다.

버튼 이벤트인 press와는달리 mouseDown은특정버튼의히트영역과는상관

이 없다. [예제 10-8]에서 볼 수 있듯이 mouseUp과 mouseMove 이벤트, 그리고

Mouse.hide() 메소드와 mouseDown 이벤트를같이사용하면사용자정의마우스

포인터를구현할수있다.

mouseUp

mouseUp 이벤트는 mouseDown 이벤트와 반대되는 이벤트이다. 이 이벤트는

마우스 포인터가 스테이지의 임의의 지점에 있을 때 마우스의 주 버튼에서 손가락

을 뗄 때마다 발생한다. mouseDown과 마찬가지로 이벤트 결과를 나타내려면 마

우스버튼에서손가락을뗄때스테이지에 mouseUp 핸들러가들어있는클립이있

어야한다. mouseUp, mouseDown, mouseMove 이벤트를이용하면 (버튼과마찬

가지로) 마우스 포인터의 모양에 영향을 끼치지 않고 마우스의 움직임에 반응하는

다양한프로그램을만들수있다.

mouseMove

mouseMove 이벤트를 이용하면 마우스 포인터의 위치 변화를 알아낼 수 있다.

마우스가 움직일 때마다 프로세서에서 새로운 이벤트를 발생시킬 수 있는 최대 속

도로 mouseMove 이벤트를발송한다. mouseMove 이벤트결과를나타내려면마우

스가움직일때스테이지에 mouseMove 핸들러가들어있는클립이있어야한다.

mouseMove 이벤트는 정지 상태에 있는 애플리케이션을 다시 시작하거나 마우

스가 움직인 경로를 표시하거나 사용자 정의 포인터를 만드는 경우에 유용하다. 이

와관련된코드는 [예제 10-8]에나와있다.

keyDown

keyDown과 keyUp 이벤트는 mouseDown과 mouseUp 이벤트를 키보드에서

적용한 것과 비슷하다. 이 두 이벤트를 이용하면 키보드를 통해 사용자와 상호작용

하는코드를만들수있다. keyDown 이벤트는키보드에있는키를누를때발생한

다. 키를 계속 누르고 있으면 OS와 키보드 설정에 따라 조금씩 다르긴 하지만

keyDown 이벤트가 계속해서 발생할 수 있다. keyPress 버튼 이벤트와는 달리

keyDown 클립이벤트는특정키만이아니라임의의키를누를때도발생한다.

keyDown 이벤트를 감지하려면 키를 눌렀을 때 스테이지에 keyDown 이벤트

핸들러가들어있는무비클립이있어야한다. 아래코드는키가눌린것을감지하는

이벤트핸들러이다.

onClipEvent (keyDown) {

trace("Some key was pressed");

}

keyDown 핸들러에서는어떤키가눌렸는지알수없다. 사용자가아무키나누

르면다음단계로넘어가도록하는경우에는어떤키를눌러도상관없다. 하지만보

통특정키에고유기능을부여하는경우가많다. 예를 들어여러키를이용하여우

주선을움직이는방향을조절하는것도그와같은경우에해당한다.

keyDown 이벤트가발생했을때어떤키가눌렸는지알아내려면키보드상태를

알려주는내장객체인 Key 객체를이용하면된다. 우리가필요로하는정보의유형

은우리가원하는상호작용에따라달라진다. 예를들어게임의경우에는여러개의

키를 연속으로 누르더라도 연속적으로, 그리고 즉각적으로 반응해야 한다. 반면에

위치를이동하기위한인터페이스에서는하나의키만을감지해도된다(예를들면슬

라이드쇼프레젠테이션의경우스페이스바와같은것).

Key 객체에서는사용자가마지막으로누른키와현재누르고있는키에대한정

보를 구할 수 있다. 키보드 상태는 Key 객체에 있는 네 개의 메소드를 이용하여 확

인할수있다.

Key.getCode()         // 마지막으로 누른 키의 키코드 (10진수)

Key.getAscii()        // 마지막으로 누른 키의 ASCII 값 (10진수)

Key.isDown(keycode)   // 주어진 키를 누르고 있는 상태이면 true를 리턴한다.

Key.isToggled(keycode)// Caps Lock이나 Num Lock이 걸려 있는지 확인한다.

[예제 10-5]는마지막으로누른키의 ASCII 값을알려주는 keyDown 핸들러이다.

[예제 10-5] 마지막으로 누른 키 확인

onClipEvent (keyDown) {

// 마지막으로 누른 키의 ASCII 값을 구해서 문자로 변환한다.

lastKeyPressed = String.fromCharCode(Key.getAscii());

trace("You pressed the '" + lastKeyPressed + "' key.");

}

[예제 10-6]은 마지막으로누른키가위쪽화살표인지확인하는 keyDown 핸들

러이다.

[예제 10-6] 위쪽 화살표 키 감지

onClipEvent (keyDown) {

// 마지막으로 누른 키가 위쪽 화살표 키인지 확인한다.

// 위쪽 화살표는 Key.UP 속성으로 표기한다.

if (Key.getCode() == Key.UP) {

trace("The up arrow was the last key depressed");

}

}

키보드상태를알아내는방법에는여러가지가있으므로용도에가장적합한방

법을선택하여사용하면된다. 예를들어 Key.getAscii() 메소드는마지막으로누른

키와 연결된 문자에 해당하는 ASCII 값을 리턴하는데, 각 언어별 키보드 배치에 따

라 그 결과가 조금씩 다를 수 있다(영문 키보드의 경우에는 키보드에 있는 문자와 숫

자의 배치가 표준으로 정해져 있다). 반면에 Key.getCode() 메소드에서는 키보드에

있는특정키(그키의문자가아닌키자체)와연결된값을리턴한다. 그키가나타내

는 문자와는 상관없이 서로 붙어 있는 키 네 개를 이용하여 위치를 이동하는 데 사

용하는 것처럼 전 세계 어느 곳에서나, 또는 어떤 플랫폼에서나 실행시킬 수 있는

프로그램을 만들 때는 Key.getCode()를 사용하는 것이 더 낫다. 자세한 정보를 원

한다면 3부에있는‘Key 객체’를읽어보기바란다.

온라인 코드 창고에서 keyDown과 keyUp을 이용하는 .fla 샘플 파일을 다운로

드할수있다.

(웹 브라우저의플러그인이아닌) 독립적인 플래시 플레이어에있는메뉴명령어

(Open, Close, Fullscreen과 같은 것)가 나타나지 않도록 하려면 무비 시작 부분에

다음과같은코드를추가하면된다.

fscommand("trapallkeys", "true");

위와 같은 명령을 사용하면 Projector에서 Esc 키를 눌러도 전체화면 모드를 빠

져나갈 수가 없다. Projector에서 Esc 키를 감지하려면 다음과 같은 코드를 사용하

면된다.

onClipEvent (keyDown) {

if (Key.getCode() == Key.ESCAPE) {

// Esc 키를 눌렀을 때 실행할 코드
}

}

하지만모든브라우저에서 Esc 키를감지할수있는것은아니다. 게다가 Alt 키

또는윈도우의 Alt-Tab, Ctrl-Alt-Del 키를사용하지못하게하는방법도없다.

탭 키가눌리는것을감지하려면다음과같은핸들러가들어있는버튼을만들면

된다.

on (keyPress "<Tab>") {

// 탭 키에 반응한다.

}

독립적인 플레이어에서는 다음과 같이 클립 이벤트 핸들러를 이용하여 탭 키를

감지할수도있다.

onClipEvent (keyDown) {

if (Key.getCode() == Key.TAB) {

// 탭 키에 반응한다.

}

}

어떤 브라우저에서는 탭 키를 감지할 때 keyPress 버튼 이벤트만 사용할 수 있

는경우도있고 keyPress 버튼이벤트와 keyUp 클립이벤트를결합시켜야하는수

도있다. 다음예에서는우선 keyPress를이용하여탭키를감지하고 keyUp 핸들러

에서탭키를눌렀을때실행할코드를처리한다. 인터넷익스플로러에서는탭키의

Key.getCode() 값이탭키에서손가락을뗄때만설정되기때문에, keyDown 이벤

트는사용하지않는다.

// 메메인인 타타임임라라인인에에 있있는는 버버튼튼의의 코코드드
on (keyPress "<Tab>") {

// 별 필요 없는 변수를 설정한다.

foo = 0;

}

// 메메인인 타타임임라라인인에에 있있는는 무무비비 클클립립의의 코코드드
onClipEvent (keyUp) {

if (Key.getCode() == Key.TAB) {

// _levle0에 있는 myTextField로 커서를 옮긴다.

Selection.setFocus("_level0.myTextField");

}

}

보통 폼에 있는 특정 텍스트 필드의 입력 위치로 움직일 때 탭 키를 이용한다.

자세한 내용을 원한다면 3부의‘Selection.setFocus() 메소드’에 있는 예제를 참조

하기바란다.

Ctrl-F와 같은 단축키 형태의 키 조합을 감지하려면 enterFrame 핸들러와

Key.isDown() 메소드를이용하면된다.

onClipEvent (enterFrame) {

if (Key.isDown(Key.CONTROL) && Key.isDown(70)) {

// Ctrl-F에 반응하는 부분
}

}

엔터키를감지하려면다음과같이버튼핸들러를사용하거나

on (keyPress "<Enter>") {

// 엔터 키에 반응한다(예: 폼 전송).

}

아니면다음과같이 keyDown 핸들러를이용하면된다.

onClipEvent (keyDown) {

if (Key.getCode() == Key.ENTER) {

// 엔터 키에 반응한다(예: 폼 전송).

}

}

기능 키(F1, F2,...)나 숫자 키패드에 있는 키와 같이 다른 특수 키를 감지하는

방법은 3부의‘Key 객체’와‘Key.getCode() 메소드’에서자세히다룬다.

keyUp

keyUp 이벤트는 키를 눌렀다가 뗄 때 발생한다. keyUp 이벤트는 게임 프로그

래밍에서 필수적인 요소이다. keyDown 이벤트를 이용해서 켰던 어떤 기능을

keyUp 이벤트를이용하여끄는경우가많기때문이다(대표적인예로는우주선게임

에서 특정 키를 누르고 있는 동안 추진력을 강화시키는 경우를 생각할 수 있다). 플래

시저작도구에서는스페이스바를잠시누르고있으면 Hand 도구로전환되고스페

이스바에서손을떼면다시원래사용하던도구로돌아간다. 이런식으로임시메뉴

와같은것을애플리케이션에서화면에표시했다가다시사라지게할수도있다.

keyDown 이벤트와 마찬가지로 keyUp 이벤트에서도 유용한 정보를 알아내려

면 Key 객체를사용해야한다.

onClipEvent (keyUp) {

if (!Key.isDown(Key.LEFT)) {

trace("The left arrow is not depressed");

}

}

Key.isDown() 메소드를 이용하면 언제나 모든 키의 상태를 알 수 있으므로

enterFrame 이벤트 루프를 이용하여 특정 키를 눌렀는지 알 수 있다. 하지만 키보

드를 계속해서 감시하는 것은 keyDown 이벤트가 이벤트 핸들러를 실행시키는 것

처럼 사용자가 어떤 키를 눌렀다는 것을 알아낼 때까지 기다리는 것보다는 비효율

적이다.

결국 어떤 방법을 사용할지는 만들고 있는 시스템의 유형에 따라 달라진다. 게

임과 같이 계속해서 변화하는 시스템에서는 매 프레임마다 메인 게임 루프를 돌기

때문에키보드를계속해서감시하는것이좋다. 즉 루프를돌리면서 Key 객체를확

인해주기만하면된다. 예를들면다음과같다.

// 비비어어있있는는 클클립립의의 코코드드
// 이렇게 하면 게임 프로세스가 계속 실행된다.

onClipEvent (enterFrame) {

_root.mainLoop();

}

// 메메인인 타타임임라라인인에에 있있는는 게게임임의의 핵핵심심 코코드드
// 프레임마다 한 번씩 실행된다.

function mainLoop () {

if (Key.isDown(Key.LEFT)) {

trace("The left arrow is depressed");

// 우주선을 왼쪽으로 회전시킨다.

}

// 다른 키를 확인해보고 게임에 필요한 작업을 처리한다.

}

하지만 정적인 인터페이스 환경에서는 특정한 키 조합(여러 키를 한꺼번에 누르

는것)을 감지하는경우를제외하면, enterFrame 루프를이용하여키보드를체크하

지 않아도 된다. 매번 키를 누를 때, 키에서 손을 뗄 때마다 한 번씩 실행되는

keyDown 및 keyUp 이벤트 핸들러만 사용해도 된다. keyUp과 keyDown 이벤트

핸들러를사용할때는어떤주어진시각에키가눌린상태인지몰라도상관없다. 따

라서 사용자가 프레임 사이에서 키에서 손을 떼더라도 키가 눌리는 것을 정확하게

감지할수있으므로키를한번만눌렀을때도같은키를두번조사할필요가없다.

하지만 어떤 경우에든 keyDown이나 keyUp 이벤트 핸들러에서 마지막으로 누른

키를확인하기위해 Key.getCode() 또는 Key.getAscii() 메소드를사용한다.

실행 순서
어떤무비에서는코드가여러타임라인과여러클립이벤트핸들러에퍼져있는

경우도있다. 따라서한프레임에서여러코드블록을실행해야하는경우도그리드

물지는 않다(어떤 코드는 이벤트 핸들러에 있고 어떤 코드는 클립 타임라인에 있는 프

레임에, 어떤 코드는 플레이어에 있는 문서의 메인 타임라인에 있을 수도 있다). 이런

경우에는여러코드가실행되는순서가꽤복잡해지고그순서가프로그램의기능에

큰영향을미칠수있다. 무비에있는여러타임라인과비교해서이벤트핸들러가실

행되는순서를확실하게알아두면예기치못한상황이발생할가능성도줄어들고코

드를원하는대로실행시킬수있다.

비동기 이벤트 핸들러는 무비의 타임라인에 있는 코드와는 독립적으로 실행된

다. 예를들어버튼이벤트핸들러는그핸들러에서처리하는이벤트가발생하면바

로실행된다. 또한 mouseDown, mouseUp, mouseMove, keyDown, keyUp 이벤

트도이벤트가발생하는대로바로실행된다.

하지만 무비 재생 이벤트는 [표 10-3]에 나온 것처럼 무비가 진행됨에 따라 순

서대로실행된다.

[표 10-3] 무비 클립 이벤트 핸들러 실행 순서

이벤트 핸들러 실행되는 때

load 클립이 스테이지에 나타난 첫 번째 프레임에서 부모 타임라인의 코드가 실
행된 후에 실행된다. 하지만 프레임이 렌더링되기 전에, 그리고 클립 내부
코드가 실행되기 전에 실행된다.

unload 클립이 스테이지에서 없어진 후 첫 번째 프레임에서 부모 타임라인의 코드
가 실행되기 전에 실행된다.

enterFrame 클립이 스테이지에 있는 동안 두 번째 이후의 모든 프레임에서 실행된다. 부
모 타임라인의 코드나 클립 내부 코드가 실행되기 전에 실행된다.

data 클립에서 데이터를 받으면 어떤 프레임에서라도 실행된다. data 이벤트가 발
생하면 클립 내부 코드나 enterFrame 코드가 실행되기 전에 실행된다.

[표 10-3]에 나온 규칙은 실질적인 예제와 함께 알아보면 더 쉽게 이해할 수 있

다. 메인 타임라인에 네 개의 키프레임이 있는 한 개의 레이어를 가지고 있는 무비

가있다고가정하자. 각 키프레임에코드를추가하고 1번프레임부터 3번프레임에

걸쳐서무비클립이들어가는두번째레이어를만든다(4번프레임은포함하지않는

다). 그리고 이 클립에 load, enterFrame, unload 핸들러를 추가한다. 마지막으로

클립 안에 각각 코드 블록을 포함하고 있는 세 개의 키프레임을 추가한다. 이러한

무비를만들면 [그림 10-2]와같은모양이된다.

무비를재생하면다음과같은순서로실행된다.

========1번 프레임=======

1) 메인 타임라인 코드가 실행된다.

2) load 핸들러가 실행된다.

3) 1번 프레임의 클립 내부 코드가 실행된다.

========2번 프레임=======

1) enterFrame 핸들러가 실행된다.

2) 2번 프레임의 클립 내부 코드가 실행된다.

3) 메인 타임라인 코드가 실행된다.

========3번 프레임=======

1) enterFrame 핸들러가 실행된다.

2) 3번 프레임의 클립 내부 코드가 실행된다.

3) 메인 타임라인 코드가 실행된다.

========4번 프레임=======

1) unload 핸들러가 실행된다.

2) 메인 타임라인 코드가 실행된다.

이 무비 예제의 실행 순서를 보면 이벤트 핸들러를 이용한 코드를 만들 때 지켜

야할중요한규칙을이해할수있다.

䤎 load 핸들러에 있는 코드는 내부 클립 코드보다 먼저 실행되므로 어떤 클립

의 1번 프레임에서 바로 사용할 변수를 초기화할 때는 load 핸들러를 사용

할수있다.

䤎 프레임에 무비 클립의 인스턴스가 만들어지기 전에 그 프레임에 있는 코드

가 실행된다. 따라서 어떤 클립 다음에 나오는 프레임이 처음으로 스테이지

에 나타나기 전에는 그러한 변수를 클립의 load 핸들러에서 선언한 경우에

도 무비 클립에 있는 사용자 정의 변수나 함수를 그 부모 타임라인에 있는

코드에서는사용할수없다.

䤎 enterFrame 이벤트는 load나 unload 이벤트와는 달리 같은 프레임에서 두

번 이상 발생하지 않는다. 어떤 클립이 스테이지에 나타났다가 바로 사라지

는경우에는 load 또는 unload 이벤트를대신사용해야한다.

䤎 각 프레임에서 클립의 enterFrame 핸들러에 있는 코드는 클립의 부모 타임

라인에 있는 코드가 실행되기 전에 실행된다. 따라서 enterFrame 핸들러를

이용하면 하나의 프레임에서 그 클립의 부모 타임라인의 속성을 변경하고

새로운값을그타임라인의코드에서곧장사용할수있다.

클립 이벤트 핸들러 복사
duplicateMovieClip() 함수를이용하여무비클립을복사하면무비클립이벤트

핸들러도함께복사된다. 예를들어다음과같이 load 이벤트핸들러를가지고있는

square라는무비클립이스테이지에있다고가정하자.

onClipEvent (load) {

trace("movie loaded");

}

square를복사하여 square2라는클립을만들면어떻게될까?

square.duplicateMovieClip("square2", 0);

square를 복사할 때 load 핸들러도 square2로 복사되기 때문에, square2가 생기

면 load 핸들러가 실행되어 Output 창에“movie is loaded”라는 내용이 출력된다.

이와같이핸들러가자동으로복사되는것을이용하면매우강력한기능을가진재귀

함수를만들수있다. [예제 10-2]에이러한기능을활용한간단한예가나와있다.

updateAfterEvent를 이용한 화면 갱신
조금전에‘실행순서’절에서배웠듯이 mouseDown, mouseUp, mouseMove,

keyDown, keyUp 이벤트 핸들러는 이벤트가 발생하면 즉시 실행된다. 심지어 프

레임을렌더링하는도중에이벤트가발생하더라도이벤트핸들러는바로실행된다.

이처럼즉시실행되기때문에반응성이뛰어날수도있지만상황에따라엉뚱한

결과가나올수도있다. mouseDown, mouseUp, mouseMove, keyDown, keyUp

이벤트핸들러때문에생기는시각적인결과는다음프레임을렌더링하기전까지는

렌더링되지 않는다. 이러한 사실을 확인해 보기 위해 프레임 속도가 초당 1 프레임

인한프레임짜리무비를만들어서다음과같은코드가들어있는무비클립을스테

이지에넣어보자.

onClipEvent (mouseDown) {

_x += 2;

}

그리고나서무비를재생시키고최대한빨리마우스를클릭해보자. 사용자가클

릭한 것이 모두 기록되긴 하지만 무비 클립은 1초에 한 번씩만 움직인다는 것을 알

수있다. 따라서프레임사이에서여섯번을클릭하면다음프레임이화면에나타날

때 클립이 한꺼번에 12픽셀을 움직이고, 세 번 클릭한다면 한꺼번에 6픽셀을 움직

인다. mouseDown 이벤트 핸들러가 실행될 때마다 그 내용은 기록이 되지만 다음

프레임이 렌더링되어야 그 결과가 화면에 출력된다. 이러한 특성 때문에 원하는 움

직임을제대로구현하지못할수도있다.

다행히도다음프레임이나올때까지기다리지않고도사용자입력이벤트핸들

러에서일어나는시각적인변화를바로렌더링할수있는방법이있다. 그냥다음과

같이이벤트핸들러안에서 updateAfterEvent() 함수를사용하면된다.

onClipEvent (mouseDown) {

_x += 2;

updateAfterEvent();

}

updateAfterEvent() 함수는 mouseDown, mouseUp, mouseMove, keyDown,

keyUp 이벤트 핸들러에서만사용할수있다. 사용자 입력과 관련된 부드럽고즉각

적인시각효과를만들때에는이함수가필수적이다. [예제 10-8]에서는사용자정

의 포인터가 부드럽게 움직일 수 있도록 updateAfterEvent()를 이용한다. 하지만

버튼이벤트에서는 updateAfterEvent() 함수를사용하지않아도된다. 버튼은프레

임중간에서도원래업데이트되기때문이다.

코드 재사용
버튼이벤트와무비클립이벤트를사용할때는 9장에서배운코드집중화원칙

을잊지말도록하자. 언제나불필요하게무비의여러요소에서코드를중복하여만

들거나코드가섞이지않도록주의해야한다. 프로그래밍을하다가두개이상의이

벤트핸들러에똑같은코드를입력한다면그코드를객체에바로집어넣지않는것

이좋다. 코드를일반화시켜객체에서빼내고그코드를무비코드를모아두는다른

곳에넣는것이좋다(보통메인타임라인에코드를넣는것이좋다).

버튼이나클립핸들러에선언문을직접집어넣는것은그다지좋은방법이아니

다. 코드를 함수로 만들고 그 함수를 핸들러에서 부르면 코드를 재사용하기도 좋고

나중에 그 코드를 찾을 때도 훨씬 편리하다. 특히 버튼의 경우에는 이렇게 하는 것

이매우유용하다(필자는버튼에함수호출선언문외에는다른선언문을거의집어넣

지않는다). 무비클립의경우에는조금더신중하게판단해야한다. 상황에따라클

립에코드를직접집어넣으면더깔끔하고완벽한코드구조를만드는데도움이될

수도있기때문이다. 자신의상황과능력에맞게적당한균형을찾을때까지다양한

접근법을 시도해 보는 것이 좋다. 코드 중복 및 재사용 문제를 염두에 두고 프로그

래밍을하면큰도움이될것이다.

코드를직접버튼에추가하는것과버튼에서함수를호출하는것사이의차이점

은 9장의‘코드집중화’예제에서쉽게알수있다.

동적 무비 클립 이벤트 핸들러
이장의맨앞부분에서플래시에는두가지종류의이벤트(무비클립이나버튼에

들어가는이벤트와 XML이나 XMLSocket과같은다른데이터객체에들어가는이벤트)

가있다는것을배웠다. 데이터객체에대한이벤트핸들러를만들려면핸들러함수

이름을 객체의 속성에 대입해야 한다. 함수를 동적으로 추가하는 방법을 다시 떠올

려보자.

myXMLDoc.onLoad = function () { trace("all done loading!"); };

동적으로함수를대입하면무비를재생하는동안핸들러의기능을변경할수있

다. 핸들러의 기능을 변경하려면 다음과 같이 핸들러 속성에 새로운 함수를 대입하

기만하면된다.

myXMLDoc.onLoad = function () { gotoAndPlay("displayData"); };

또는다음과같이핸들러를없앨수도있다.

myXMLDoc.onLoad = function () { return; };

하지만 안타깝게도 무비 클립이나 버튼의 이벤트는 그다지 융통성이 없어서 무

비를재생하는도중에변경하거나제거할수없다. 게다가무비클립이벤트핸들러

는어떤무비의메인무비타임라인에추가할수없다. 즉무비의 _root 클립에는이

벤트핸들러를추가할수가없다.

이러한 한계를 극복하기 위해(enterFrame 이벤트와 사용자 입력 이벤트의 경우

에) 동적으로이벤트핸들러를제거하거나변경하는것과비슷한효과를볼수있도

록 비어있는 무비 클립을 이용한다. 비어있는 무비 클립을 이용하면 _root 레벨의

이벤트와 비슷한 기능을 구현할 수 있다. 이러한 테크닉은 사실 8장에서 다음과 같

은방법으로이벤트루프를만들때이미배웠다.

1. process라는이름의비어있는무비클립을만든다.

2. process 안에 eventClip이라는비어있는클립을하나더만든다.

3. eventClip에 원하는 이벤트 핸들러를 추가한다. eventClip의 핸들러에 있는

코드에서는 다음과 같이 process 클립의 호스트 타임라인을 대상으로 잡아

야한다.

onClipEvent (mouseMove) {

_ parent._ parent.doSomeFunction();

}

4. process를 attachMovie() 함수에서 사용할 수 있도록 Library에서 process

를 선택하고 Options → Linkage를 선택한다. Linkage를 Export This

Symbol로설정하고적당한인식자(예: “mouseMoveProcess”)를할당한다.

5. 마지막으로 이벤트 핸들러와 연결하려면 attachMovie()를 이용하여

process 클립을적당한타임라인에추가한다.

6. 핸들러와 이벤트의연결을 끊고 싶다면 removeMovieClip() 함수를 이용하

여 process 클립을제거한다.

enterFrame 이벤트에서 이러한 기법을 사용하는 법은 8장의‘플래시 5 클립 이

벤트루프’에자세히설명되어있다.

이벤트 핸들러 응용
몇가지실전예제를통해액션스크립트이벤트와이벤트핸들러에관한설명을

마칠까한다. 여기에나오는애플리케이션은매우간단하긴하지만, 이 예제들을통

해 이벤트 기반 프로그래밍의 폭넓은 응용범위에 대해 감을 잡을 수 있을 것이다.

마지막두예제는온라인코드창고에서다운로드할수있다.

[예제 10-7]은클립을줄였다키웠다하는예제이다.

[예제 10-7] 무비 클립 크기를 변동시키는 예제

onClipEvent (load) {

var shrinking = false;

var maxHeight = 300;

var minHeight = 30;

}

onClipEvent (enterFrame) {

if (_height < maxHeight && shrinking == false) {

_height += 10;

_width  += 10;

} else {

shrinking = true;

}

if (shrinking == true) {

if (_height > minHeight) {

_height -= 10;

_width  -= 10;

} else {

shrinking = false;

_height += 10;     // 한 사이클을 건너뛰지 않도록
_width  += 10;     // 값을 증가시킨다.

}

}

}

[예제 10-8]에서는 보통 쓰이는 시스템 포인터를 숨기고 마우스의 위치를 클립

이 따라다니도록 만들어서 사용자 정의 마우스 포인터와 같은 효과를 보여준다. 이

예제에서 mouseDown과 mouseUp 핸들러에서는마우스포인터의크기를약간바

꿔서마우스가클릭되는것을알려준다.

[예제 10-8] 사용자 정의 마우스 포인터

onClipEvent (load) {

Mouse.hide();

}

onClipEvent (mouseMove) {

_x = _root._xmouse;

_ y = _root._ymouse;

updateAfterEvent();

}

onClipEvent (mouseDown) {

_width  *= .5;

_height *= .5;

updateAfterEvent();

}

onClipEvent (mouseUp) {

_width  *= 2;

_height *= 2;

updateAfterEvent();

}

마지막으로 [예제 10-9]는 액션스크립트 무비 클립 이벤트 핸들러의 강력한 기

능을보여주는예로, 무비클립을사용자정의버튼으로바꾸는프로그램이다. 여기

서는 mouseMove를 이용하여 마우스가 움직이는 것을 확인하고 mouseDown과

mouseUp을 이용하여 버튼을 클릭하는 것을 감지하고 hitTest() 함수를 이용하여

간단하게 마우스의 위치를 파악한다. 이 예제에서는 핸들러를 포함하고 있는 클립

에 up, down, over(각각 버튼이눌리지않은상태, 눌린 상태, 버튼 위에마우스포인

터가있는상태를나타냄)라는레이블을가진세개의키프레임이있다고가정한다.

[예제 10-9] 무비 클립 버튼

onClipEvent (load) {

stop();

}

onClipEvent (mouseMove) {

if (hitTest(_root._xmouse, _root._ymouse, true) && !buttonDown) {

this.gotoAndStop("over");

} else if (!hitTest(_root._xmouse, _root._ymouse, true) && !buttonDown)

{

this.gotoAndStop("up");

}

updateAfterEvent();

}

onClipEvent (mouseDown) {

if (hitTest(_root._xmouse, _root._ymouse, true)) {

buttonDown = true;

this.gotoAndStop("down");

}

updateAfterEvent();

}

onClipEvent (mouseUp) {

buttonDown = false;

if (!hitTest(_root._xmouse, _root._ymouse, true)) {

this.gotoAndStop("up");

} else {

this.gotoAndStop("over");

}

{_Index {_EndRange_}event handlers:applications of_}

updateAfterEvent();

}

앞으로 배울 내용
선언문, 연산자, 함수, 그리고이벤트및이벤트핸들러까지배웠으니이제액션

스크립트의 내부 도구 작동법을 모두 배운 셈이다. 액션스크립트를 완벽하게 이해

할수있도록앞으로세장에걸쳐아주중요한데이터형인배열, 객체, 무비클립에

대해살펴볼것이다.

